Appendix A2. Converting from 3-year to 1-year transitions.
We converted 3y transition matrices to 1y transition matrices in two stages. First, we used an algorithm developed in MATLAB by Chhatwal and colleagues (2016) to get an initial approximation of the stochastic cube root by eigendecomposition, or when eigendecomposition did not produce a stochastic root, by minimizing the distance between the cube of a stochastic root and the original 3y matrix. For each matrix, we calculated the percentage error of the approximation by comparing the cube of the approximate matrix to the original 3y matrix (Chhatwal et al., 2016). Following Chhatwal and colleagues (2016), we calculated the percentage error as the Frobenius norm of the difference between 1-year (1y) matrix cubed and the 3y matrix, divided by the Frobenius norm of the 3y matrix, or [image: image2.png]11y ~3yll
113yl



. Because the average percentage error across our 24 matrices (20%) greatly exceeded the 3% error achieved by Chhatwal and colleagues (2016) in their example, we sought to improve our approximate cube roots. We developed a simple “hill climber” algorithm in Stata that applied slight modifications to each of the matrices and chose the modification that produced the smallest percentage error (Russell and Norvig, 2016). Specifically, the algorithm evaluated the impact on percentage error of adding and subtracting 0.001 to each element of the matrix and chose the value that minimized the percentage error. The algorithm repeated this process until no change meaningfully improved the percentage error (Vanni et al., 2011). The result of this second stage was a smaller but still large average percentage error of 12%. This process generated our set of short-term transition matrices, TM1. The hill climber code is shown below.

Hill Climber Code (Stata):
*Percentage Error formula (from Chhatwal et al., 2016)

   mata

      void err_meas(string scalar A, string scalar B) {

         Amat=st_matrix(A)

         Bmat=st_matrix(B)

         result = ((sum((Bmat:-(Amat*Amat*Amat)):^2))^(.5))/(sum(Bmat:^2))^(.5)

         st_matrix("result", result)

      }

   end

   /*Hill Climber*/

   mat input prepost = ()

   foreach m in age3_fem1_raeth4 age3_fem1_raeth3 age3_fem1_raeth2 age3_fem1_raeth1    age3_fem0_raeth4 age3_fem0_raeth3 age3_fem0_raeth2 age3_fem0_raeth1 age2_fem1_raeth4 age2_fem1_raeth3 age2_fem1_raeth2 age2_fem1_raeth1 age2_fem0_raeth4 age2_fem0_raeth3 age2_fem0_raeth2 age2_fem0_raeth1 age1_fem1_raeth4 age1_fem1_raeth3 age1_fem1_raeth2 age1_fem1_raeth1 age1_fem0_raeth4 age1_fem0_raeth3 age1_fem0_raeth2 age1_fem0_raeth1 {

      *starting percentage error

      mata: err_meas("y1`m'", "y3`m'")

      local error = result[1,1]

      local origerr = `error'

      mat current = y1`m'

      mat input hc_`m' = (., ., ., `origerr')

      local besterror = `error'

      local stop = 0

      while `stop' == 0 {

         *Check all elements and add/subtract, choose the best

         forvalues r = 1/5 {

            forvalues c = 1/5 {

               foreach a in -1 1 {

                  mat tmp = current

                  local origval = tmp[`r',`c']

                  if `a' == 1 { /*Add*/

                     local change = min(.99999-current[`r',`c'], .001)

                  }

                  else { /*Subtract*/

                     local change = min(current[`r',`c']-.00001, .001)

                  }

                  mat tmp[`r',`c'] = tmp[`r',`c']+`change'*`a'

                  local newval = tmp[`r', `c']

                  forvalues y = 1/5 {

                     if `y' != `c' {

                        mat tmp[`r',`y'] = tmp[`r',`y']/(1-`origval')*(1-`newval')

                     }

                  }

                  mata: err_meas("tmp", "y3`m'")

                  local tmperror = result[1,1]

                  if `tmperror' < `besterror' {

                     local bestrow = `r'

                     local bestcol = `c'

                     local bestop = `a'

                     local besterror = `tmperror'

                  }

               }

            }

         }

         *Now apply the best one if it's better (by .000001). If it's not better by that threshold, stop

         if (`besterror'+0.000001) < `error' {

            local origval = current[`bestrow',`bestcol']

            if `bestop' == 1 { /*Add*/

               local change = min(.99999-current[`bestrow',`bestcol'], .001)

            }

            else { /*Subtract*/

               local change = min(current[`bestrow',`bestcol']-.00001, .001)

            }

            mat current[`bestrow',`bestcol'] = current[`bestrow',`bestcol']+`change'*`bestop'

            local newval = current[`bestrow', `bestcol']

            forvalues y = 1/5 {

               if `y' != `bestcol' {

                  mat current[`bestrow',`y'] = current[`bestrow',`y']/(1-`origval')*(1-`newval')

               }

            }

            mata: err_meas("current", "y3`m'")

            local error = result[1,1]

            local besterror = `error'

            mat hc_`m' = hc_`m'\(`bestrow', `bestcol', `bestop', `error')

         }

         else {

            local stop = 1

         }

         di "Row: `bestrow'; Column: `bestcol'; Operation: `bestop'; Error: `error'"

      }
